
Chapter 4. Integration of Functions

4.0 Introduction

Numerical integration, which is also called quadrature, has a history extending
back to the invention of calculus and before. The fact that integrals of elementary
functions could not, in general, be computed analytically, while derivatives could
be, served to give the field a certain panache, and to set it a cut above the arithmetic
drudgery of numerical analysis during the whole of the 18th and 19th centuries.

With the invention of automatic computing, quadrature became just one numer-
ical task among many, and not a very interesting one at that. Automatic computing,
even the most primitive sort involving desk calculators and rooms full of “computers”
(that were, until the 1950s, people rather than machines), opened to feasibility the
much richer field of numerical integration of differential equations. Quadrature is
merely the simplest special case: The evaluation of the integral

I =
∫ b

a

f(x)dx (4.0.1)

is precisely equivalent to solving for the value I ≡ y(b) the differential equation

dy

dx
= f(x) (4.0.2)

with the boundary condition

y(a) = 0 (4.0.3)

Chapter 16 of this book deals with the numerical integration of differential
equations. In that chapter, much emphasis is given to the concept of “variable” or
“adaptive” choices of stepsize. We will not, therefore, develop that material here.
If the function that you propose to integrate is sharply concentrated in one or more
peaks, or if its shape is not readily characterized by a single length-scale, then it
is likely that you should cast the problem in the form of (4.0.2)–(4.0.3) and use
the methods of Chapter 16.

The quadrature methods in this chapter are based, in one way or another, on the
obvious device of adding up the value of the integrand at a sequence of abscissas
within the range of integration. The game is to obtain the integral as accurately
as possible with the smallest number of function evaluations of the integrand. Just
as in the case of interpolation (Chapter 3), one has the freedom to choose methods

of various orders, with higher order sometimes, but not always, giving higher
accuracy. “Romberg integration,” which is discussed in §4.3, is a general formalism
for making use of integration methods of a variety of different orders, and we
recommend it highly.

Apart from the methods of this chapter and of Chapter 16, there are yet
other methods for obtaining integrals. One important class is based on function
approximation. We discuss explicitly the integration of functions by Chebyshev
approximation (“Clenshaw-Curtis” quadrature) in §5.9. Although not explicitly
discussed here, you ought to be able to figure out how to do cubic spline quadrature
using the output of the routine spline in §3.3. (Hint: Integrate equation 3.3.3
over x analytically. See [1].)

Some integrals related to Fourier transforms can be calculated using the fast
Fourier transform (FFT) algorithm. This is discussed in §13.9.

Multidimensional integrals are another whole multidimensional bag of worms.
Section 4.6 is an introductory discussion in this chapter; the important technique of
Monte-Carlo integration is treated in Chapter 7.

CITED REFERENCES AND FURTHER READING:

Carnahan, B., Luther, H.A., and Wilkes, J.O. 1969, Applied Numerical Methods (New York:
Wiley), Chapter 2.

Isaacson, E., and Keller, H.B. 1966, Analysis of Numerical Methods (New York: Wiley), Chapter 7.

Acton, F.S. 1970, Numerical Methods That Work; 1990, corrected edition (Washington: Mathe-
matical Association of America), Chapter 4.

Stoer, J., and Bulirsch, R. 1980, Introduction to Numerical Analysis (New York: Springer-Verlag),
Chapter 3.

Ralston, A., and Rabinowitz, P. 1978, A First Course in Numerical Analysis, 2nd ed. (New York:
McGraw-Hill), Chapter 4.

Dahlquist, G., and Bjorck, A. 1974, Numerical Methods (Englewood Cliffs, NJ: Prentice-Hall),
§7.4.

Kahaner, D., Moler, C., and Nash, S. 1989, Numerical Methods and Software (Englewood Cliffs,
NJ: Prentice Hall), Chapter 5.

Forsythe, G.E., Malcolm, M.A., and Moler, C.B. 1977, Computer Methods for Mathematical
Computations (Englewood Cliffs, NJ: Prentice-Hall), §5.2, p. 89. [1]

Davis, P., and Rabinowitz, P. 1984, Methods of Numerical Integration, 2nd ed. (Orlando, FL:
Academic Press).

4.1 Classical Formulas for Equally Spaced
Abscissas

Where would any book on numerical analysis be without Mr. Simpson and his
“rule”? The classical formulas for integrating a function whose value is known at
equally spaced steps have a certain elegance about them, and they are redolent with
historical association. Through them, the modern numerical analyst communes with
the spirits of his or her predecessors back across the centuries, as far as the time
of Newton, if not farther. Alas, times do change; with the exception of two of the
most modest formulas (“extended trapezoidal rule,” equation 4.1.11, and “extended



4.1 Classical Formulas for Equally Spaced Abscissas 125

S
am

ple page from
 N

U
M

E
R

IC
A

L R
E

C
IP

E
S

 IN
 F

O
R

T
R

A
N

 77: T
H

E
 A

R
T

 O
F

 S
C

IE
N

T
IF

IC
 C

O
M

P
U

T
IN

G
 (IS

B
N

 0-521-43064-X
)

C
opyright (C

) 1986-1992 by C
am

bridge U
niversity P

ress.P
rogram

s C
opyright (C

) 1986-1992 by N
um

erical R
ecipes S

oftw
are. 

P
erm

ission is granted for internet users to m
ake one paper copy for their ow

n personal use. F
urther reproduction, or any copying of m

achine-
readable files (including this one) to any server

com
puter, is strictly prohibited. T

o order N
um

erical R
ecipes books

or C
D

R
O

M
s, visit w

ebsite
http://w

w
w

.nr.com
 or call 1-800-872-7423 (N

orth A
m

erica only),or send em
ail to directcustserv@

cam
bridge.org (outside N

orth A
m

erica).

x0 xN xN + 1

open formulas use these points

closed formulas use these points

x1 x2

h

Figure 4.1.1. Quadrature formulas with equally spaced abscissas compute the integral of a function
between x0 and xN+1. Closed formulas evaluate the function on the boundary points, while open
formulas refrain from doing so (useful if the evaluation algorithm breaks down on the boundary points).

midpoint rule,” equation 4.1.19, see §4.2), the classical formulas are almost entirely
useless. They are museum pieces, but beautiful ones.

Some notation: We have a sequence of abscissas, denoted x0, x1, . . . , xN ,
xN+1 which are spaced apart by a constant step h,

xi = x0 + ih i = 0, 1, . . . , N + 1 (4.1.1)

A function f(x) has known values at the xi’s,

f(xi) ≡ fi (4.1.2)

We want to integrate the function f(x) between a lower limit a and an upper limit
b, where a and b are each equal to one or the other of the x i’s. An integration
formula that uses the value of the function at the endpoints, f(a) or f(b), is called
a closed formula. Occasionally, we want to integrate a function whose value at one
or both endpoints is difficult to compute (e.g., the computation of f goes to a limit
of zero over zero there, or worse yet has an integrable singularity there). In this
case we want an open formula, which estimates the integral using only x i’s strictly
between a and b (see Figure 4.1.1).

The basic building blocks of the classical formulas are rules for integrating a
function over a small number of intervals. As that number increases, we can find
rules that are exact for polynomials of increasingly high order. (Keep in mind that
higher order does not always imply higher accuracy in real cases.) A sequence of
such closed formulas is now given.

Closed Newton-Cotes Formulas

Trapezoidal rule:

∫ x2

x1

f(x)dx = h

[
1
2
f1 +

1
2
f2

]
+ O(h3f ′′) (4.1.3)

Here the error term O( ) signifies that the true answer differs from the estimate by
an amount that is the product of some numerical coefficient times h 3 times the value

of the function’s second derivative somewhere in the interval of integration. The
coefficient is knowable, and it can be found in all the standard references on this
subject. The point at which the second derivative is to be evaluated is, however,
unknowable. If we knew it, we could evaluate the function there and have a higher-
order method! Since the product of a knowable and an unknowable is unknowable,
we will streamline our formulas and write only O( ), instead of the coefficient.

Equation (4.1.3) is a two-point formula (x1 and x2). It is exact for polynomials
up to and including degree 1, i.e., f(x) = x. One anticipates that there is a
three-point formula exact up to polynomials of degree 2. This is true; moreover, by a
cancellation of coefficients due to left-right symmetry of the formula, the three-point
formula is exact for polynomials up to and including degree 3, i.e., f(x) = x 3:

Simpson’s rule:

∫ x3

x1

f(x)dx = h

[
1
3
f1 +

4
3
f2 +

1
3
f3

]
+ O(h5f (4)) (4.1.4)

Here f (4) means the fourth derivative of the function f evaluated at an unknown
place in the interval. Note also that the formula gives the integral over an interval
of size 2h, so the coefficients add up to 2.

There is no lucky cancellation in the four-point formula, so it is also exact for
polynomials up to and including degree 3.

Simpson’s 3
8 rule:

∫ x4

x1

f(x)dx = h

[
3
8
f1 +

9
8
f2 +

9
8
f3 +

3
8
f4

]
+ O(h5f (4)) (4.1.5)

The five-point formula again benefits from a cancellation:

Bode’s rule:

∫ x5

x1

f(x)dx = h

[
14
45

f1 +
64
45

f2 +
24
45

f3 +
64
45

f4 +
14
45

f5

]
+O(h7f (6)) (4.1.6)

This is exact for polynomials up to and including degree 5.
At this point the formulas stop being named after famous personages, so we

will not go any further. Consult [1] for additional formulas in the sequence.

Extrapolative Formulas for a Single Interval

We are going to depart from historical practice for a moment. Many texts
would give, at this point, a sequence of “Newton-Cotes Formulas of Open Type.”
Here is an example:

∫ x5

x0

f(x)dx = h

[
55
24

f1 +
5
24

f2 +
5
24

f3 +
55
24

f4

]
+ O(h5f (4))

Notice that the integral from a = x0 to b = x5 is estimated, using only the interior
points x1, x2, x3, x4. In our opinion, formulas of this type are not useful for the
reasons that (i) they cannot usefully be strung together to get “extended” rules, as we
are about to do with the closed formulas, and (ii) for all other possible uses they are
dominated by the Gaussian integration formulas which we will introduce in §4.5.

Instead of the Newton-Cotes open formulas, let us set out the formulas for
estimating the integral in the single interval from x0 to x1, using values of the
function f at x1, x2, . . . . These will be useful building blocks for the “extended”
open formulas.

∫ x1

x0

f(x)dx = h[f1] + O(h2f ′) (4.1.7)

∫ x1

x0

f(x)dx = h

[
3
2
f1 − 1

2
f2

]
+ O(h3f ′′) (4.1.8)

∫ x1

x0

f(x)dx = h

[
23
12

f1 − 16
12

f2 +
5
12

f3

]
+ O(h4f (3)) (4.1.9)

∫ x1

x0

f(x)dx = h

[
55
24

f1 − 59
24

f2 +
37
24

f3 − 9
24

f4

]
+ O(h5f (4))(4.1.10)

Perhaps a word here would be in order about how formulas like the above can
be derived. There are elegant ways, but the most straightforward is to write down the
basic form of the formula, replacing the numerical coefficients with unknowns, say
p, q, r, s. Without loss of generality take x0 = 0 and x1 = 1, so h = 1. Substitute in
turn for f(x) (and for f1, f2, f3, f4) the functions f(x) = 1, f(x) = x, f(x) = x2,
and f(x) = x3. Doing the integral in each case reduces the left-hand side to a
number, and the right-hand side to a linear equation for the unknowns p, q, r, s.
Solving the four equations produced in this way gives the coefficients.

Extended Formulas (Closed)

If we use equation (4.1.3) N − 1 times, to do the integration in the intervals
(x1, x2), (x2, x3), . . . , (xN−1, xN ), and then add the results, we obtain an “extended”
or “composite” formula for the integral from x1 to xN .

Extended trapezoidal rule:

∫ xN

x1

f(x)dx = h

[
1
2
f1 + f2 + f3+

· · · + fN−1 +
1
2
fN

]
+ O

(
(b − a)3f ′′

N2

) (4.1.11)

Here we have written the error estimate in terms of the interval b− a and the number
of points N instead of in terms of h. This is clearer, since one is usually holding
a and b fixed and wanting to know (e.g.) how much the error will be decreased

by taking twice as many steps (in this case, it is by a factor of 4). In subsequent
equations we will show only the scaling of the error term with the number of steps.

For reasons that will not become clear until §4.2, equation (4.1.11) is in fact
the most important equation in this section, the basis for most practical quadrature
schemes.

The extended formula of order 1/N 3 is:

∫ xN

x1

f(x)dx = h

[
5
12

f1 +
13
12

f2 + f3 + f4+

· · · + fN−2 +
13
12

fN−1 +
5
12

fN

]
+ O

(
1

N3

) (4.1.12)

(We will see in a moment where this comes from.)
If we apply equation (4.1.4) to successive, nonoverlapping pairs of intervals,

we get the extended Simpson’s rule:

∫ xN

x1

f(x)dx = h

[
1
3
f1 +

4
3
f2 +

2
3
f3 +

4
3
f4+

· · · + 2
3
fN−2 +

4
3
fN−1 +

1
3
fN

]
+ O

(
1

N4

) (4.1.13)

Notice that the 2/3, 4/3 alternation continues throughout the interior of the evalu-
ation. Many people believe that the wobbling alternation somehow contains deep
information about the integral of their function that is not apparent to mortal eyes.
In fact, the alternation is an artifact of using the building block (4.1.4). Another
extended formula with the same order as Simpson’s rule is

∫ xN

x1

f(x)dx = h

[
3
8
f1 +

7
6
f2 +

23
24

f3 + f4 + f5+

· · · + fN−4 + fN−3 +
23
24

fN−2 +
7
6
fN−1 +

3
8
fN

]

+ O

(
1

N4

)
(4.1.14)

This equation is constructed by fitting cubic polynomials through successive groups
of four points; we defer details to §18.3, where a similar technique is used in the
solution of integral equations. We can, however, tell you where equation (4.1.12)
came from. It is Simpson’s extended rule, averaged with a modified version of
itself in which the first and last step are done with the trapezoidal rule (4.1.3). The
trapezoidal step is two orders lower than Simpson’s rule; however, its contribution to
the integral goes down as an additional power of N (since it is used only twice, not
N times). This makes the resulting formula of degree one less than Simpson.

Extended Formulas (Open and Semi-open)

We can construct open and semi-open extended formulas by adding the closed
formulas (4.1.11)–(4.1.14), evaluated for the second and subsequent steps, to the
extrapolative open formulas for the first step, (4.1.7)–(4.1.10). As discussed
immediately above, it is consistent to use an end step that is of one order lower
than the (repeated) interior step. The resulting formulas for an interval open at
both ends are as follows:

Equations (4.1.7) and (4.1.11) give∫ xN

x1

f(x)dx = h

[
3
2
f2+f3 +f4+ · · ·+fN−2+

3
2
fN−1

]
+O

(
1

N2

)
(4.1.15)

Equations (4.1.8) and (4.1.12) give∫ xN

x1

f(x)dx = h

[
23
12

f2 +
7
12

f3 + f4 + f5+

· · · + fN−3 +
7
12

fN−2 +
23
12

fN−1

]

+ O

(
1

N3

)
(4.1.16)

Equations (4.1.9) and (4.1.13) give∫ xN

x1

f(x)dx = h

[
27
12

f2 + 0 +
13
12

f4 +
4
3
f5+

· · · + 4
3
fN−4 +

13
12

fN−3 + 0 +
27
12

fN−1

]

+ O

(
1

N4

)
(4.1.17)

The interior points alternate 4/3 and 2/3. If we want to avoid this alternation,
we can combine equations (4.1.9) and (4.1.14), giving∫ xN

x1

f(x)dx = h

[
55
24

f2 − 1
6
f3 +

11
8

f4 + f5 + f6 + f7+

· · · + fN−5 + fN−4 +
11
8

fN−3 − 1
6
fN−2 +

55
24

fN−1

]

+ O

(
1

N4

)

(4.1.18)

We should mention in passing another extended open formula, for use where
the limits of integration are located halfway between tabulated abscissas. This one is
known as the extended midpoint rule, and is accurate to the same order as (4.1.15):∫ xN

x1

f(x)dx = h[f3/2 + f5/2 + f7/2+

· · · + fN−3/2 + fN−1/2] + O

(
1

N2

) (4.1.19)

N = 1

2

3

4

(total after N = 4)

Figure 4.2.1. Sequential calls to the routine trapzd incorporate the information from previous calls and
evaluate the integrand only at those new points necessary to refine the grid. The bottom line shows the
totality of function evaluations after the fourth call. The routine qsimp, by weighting the intermediate
results, transforms the trapezoid rule into Simpson’s rule with essentially no additional overhead.

There are also formulas of higher order for this situation, but we will refrain from
giving them.

The semi-open formulas are just the obvious combinations of equations (4.1.11)–
(4.1.14) with (4.1.15)–(4.1.18), respectively. At the closed end of the integration,
use the weights from the former equations; at the open end use the weights from
the latter equations. One example should give the idea, the formula with error term
decreasing as 1/N 3 which is closed on the right and open on the left:

∫ xN

x1

f(x)dx = h

[
23
12

f2 +
7
12

f3 + f4 + f5+

· · · + fN−2 +
13
12

fN−1 +
5
12

fN

]
+ O

(
1

N3

) (4.1.20)

CITED REFERENCES AND FURTHER READING:

Abramowitz, M., and Stegun, I.A. 1964, Handbook of Mathematical Functions, Applied Mathe-
matics Series, Volume 55 (Washington: National Bureau of Standards; reprinted 1968 by
Dover Publications, New York), §25.4. [1]

Isaacson, E., and Keller, H.B. 1966, Analysis of Numerical Methods (New York: Wiley), §7.1.

4.2 Elementary Algorithms

Our starting point is equation (4.1.11), the extended trapezoidal rule. There are
two facts about the trapezoidal rule which make it the starting point for a variety of
algorithms. One fact is rather obvious, while the second is rather “deep.”

The obvious fact is that, for a fixed function f(x) to be integrated between fixed
limits a and b, one can double the number of intervals in the extended trapezoidal
rule without losing the benefit of previous work. The coarsest implementation of
the trapezoidal rule is to average the function at its endpoints a and b. The first
stage of refinement is to add to this average the value of the function at the halfway
point. The second stage of refinement is to add the values at the 1/4 and 3/4 points.
And so on (see Figure 4.2.1).

Without further ado we can write a routine with this kind of logic to it:

SUBROUTINE trapzd(func,a,b,s,n)
INTEGER n
REAL a,b,s,func
EXTERNAL func

This routine computes the nth stage of refinement of an extended trapezoidal rule. func is
input as the name of the function to be integrated between limits a and b, also input. When

called with n=1, the routine returns as s the crudest estimate of
∫ b
a

f(x)dx. Subsequent

calls with n=2,3,... (in that sequential order) will improve the accuracy of s by adding 2n-2

additional interior points. s should not be modified between sequential calls.
INTEGER it,j
REAL del,sum,tnm,x
if (n.eq.1) then

s=0.5*(b-a)*(func(a)+func(b))
else

it=2**(n-2)
tnm=it
del=(b-a)/tnm This is the spacing of the points to be added.
x=a+0.5*del
sum=0.
do 11 j=1,it

sum=sum+func(x)
x=x+del

enddo 11

s=0.5*(s+(b-a)*sum/tnm) This replaces s by its refined value.
endif
return
END

The above routine (trapzd) is a workhorse that can be harnessed in several
ways. The simplest and crudest is to integrate a function by the extended trapezoidal
rule where you know in advance (we can’t imagine how!) the number of steps you
want. If you want 2M + 1, you can accomplish this by the fragment

do 11 j=1,m+1
call trapzd(func,a,b,s,j)

enddo 11

with the answer returned as s.
Much better, of course, is to refine the trapezoidal rule until some specified

degree of accuracy has been achieved:

SUBROUTINE qtrap(func,a,b,s)
INTEGER JMAX
REAL a,b,func,s,EPS
EXTERNAL func
PARAMETER (EPS=1.e-6, JMAX=20)

C USES trapzd
Returns as s the integral of the function func from a to b. The parameters EPS can be set
to the desired fractional accuracy and JMAX so that 2 to the power JMAX-1 is the maximum
allowed number of steps. Integration is performed by the trapezoidal rule.

INTEGER j
REAL olds
olds=-1.e30 Any number that is unlikely to be the average of the function

at its endpoints will do here.do 11 j=1,JMAX
call trapzd(func,a,b,s,j)
if (j.gt.5) then Avoid spurious early convergence.

if (abs(s-olds).lt.EPS*abs(olds).or.
* (s.eq.0..and.olds.eq.0.)) return

endif
olds=s

enddo 11

pause ’too many steps in qtrap’
END

Unsophisticated as it is, routine qtrap is in fact a fairly robust way of doing
integrals of functions that are not very smooth. Increased sophistication will usually
translate into a higher-order method whose efficiency will be greater only for
sufficiently smooth integrands. qtrap is the method of choice, e.g., for an integrand
which is a function of a variable that is linearly interpolated between measured data
points. Be sure that you do not require too stringent an EPS, however: If qtrap takes
too many steps in trying to achieve your required accuracy, accumulated roundoff
errors may start increasing, and the routine may never converge. A value 10 −6

is just on the edge of trouble for most 32-bit machines; it is achievable when the
convergence is moderately rapid, but not otherwise.

We come now to the “deep” fact about the extended trapezoidal rule, equation
(4.1.11). It is this: The error of the approximation, which begins with a term of
order 1/N 2, is in fact entirely even when expressed in powers of 1/N . This follows
directly from the Euler-Maclaurin Summation Formula,

∫ xN

x1

f(x)dx = h

[
1
2
f1 + f2 + f3 + · · · + fN−1 +

1
2
fN

]

− B2h
2

2!
(f ′

N − f ′
1) − · · · − B2kh2k

(2k)!
(f (2k−1)

N − f
(2k−1)
1 ) − · · ·

(4.2.1)

Here B2k is a Bernoulli number, defined by the generating function

t

et − 1
=

∞∑
n=0

Bn
tn

n!
(4.2.2)

with the first few even values (odd values vanish except for B1 = −1/2)

B0 = 1 B2 =
1
6

B4 = − 1
30

B6 =
1
42

B8 = − 1
30

B10 =
5
66

B12 = − 691
2730

(4.2.3)

Equation (4.2.1) is not a convergent expansion, but rather only an asymptotic
expansion whose error when truncated at any point is always less than twice the
magnitude of the first neglected term. The reason that it is not convergent is that
the Bernoulli numbers become very large, e.g.,

B50 =
495057205241079648212477525

66

The key point is that only even powers of h occur in the error series of (4.2.1).
This fact is not, in general, shared by the higher-order quadrature rules in §4.1.
For example, equation (4.1.12) has an error series beginning with O(1/N 3), but
continuing with all subsequent powers of N : 1/N 4, 1/N5, etc.

Suppose we evaluate (4.1.11) with N steps, getting a result SN , and then again
with 2N steps, getting a result S2N . (This is done by any two consecutive calls of

trapzd.) The leading error term in the second evaluation will be 1/4 the size of the
error in the first evaluation. Therefore the combination

S =
4
3
S2N − 1

3
SN (4.2.4)

will cancel out the leading order error term. But there is no error term of order 1/N 3,
by (4.2.1). The surviving error is of order 1/N 4, the same as Simpson’s rule. In fact,
it should not take long for you to see that (4.2.4) is exactly Simpson’s rule (4.1.13),
alternating 2/3’s, 4/3’s, and all. This is the preferred method for evaluating that rule,
and we can write it as a routine exactly analogous to qtrap above:

SUBROUTINE qsimp(func,a,b,s)
INTEGER JMAX
REAL a,b,func,s,EPS
EXTERNAL func
PARAMETER (EPS=1.e-6, JMAX=20)

C USES trapzd
Returns as s the integral of the function func from a to b. The parameters EPS can be set
to the desired fractional accuracy and JMAX so that 2 to the power JMAX-1 is the maximum
allowed number of steps. Integration is performed by Simpson’s rule.

INTEGER j
REAL os,ost,st
ost=-1.e30
os= -1.e30
do 11 j=1,JMAX

call trapzd(func,a,b,st,j)
s=(4.*st-ost)/3. Compare equation (4.2.4), above.
if (j.gt.5) then Avoid spurious early convergence.

if (abs(s-os).lt.EPS*abs(os).or.
* (s.eq.0..and.os.eq.0.)) return

endif
os=s
ost=st

enddo 11

pause ’too many steps in qsimp’
END

The routine qsimp will in general be more efficient than qtrap (i.e., require
fewer function evaluations) when the function to be integrated has a finite 4th
derivative (i.e., a continuous 3rd derivative). The combination of qsimp and its
necessary workhorse trapzd is a good one for light-duty work.

CITED REFERENCES AND FURTHER READING:

Stoer, J., and Bulirsch, R. 1980, Introduction to Numerical Analysis (New York: Springer-Verlag),
§3.3.

Dahlquist, G., and Bjorck, A. 1974, Numerical Methods (Englewood Cliffs, NJ: Prentice-Hall),
§§7.4.1–7.4.2.

Forsythe, G.E., Malcolm, M.A., and Moler, C.B. 1977, Computer Methods for Mathematical
Computations (Englewood Cliffs, NJ: Prentice-Hall), §5.3.

4.3 Romberg Integration

We can view Romberg’s method as the natural generalization of the routine
qsimp in the last section to integration schemes that are of higher order than
Simpson’s rule. The basic idea is to use the results from k successive refinements
of the extended trapezoidal rule (implemented in trapzd) to remove all terms in
the error series up to but not including O(1/N 2k). The routine qsimp is the case
of k = 2. This is one example of a very general idea that goes by the name of
Richardson’s deferred approach to the limit: Perform some numerical algorithm for
various values of a parameter h, and then extrapolate the result to the continuum
limit h = 0.

Equation (4.2.4), which subtracts off the leading error term, is a special case of
polynomial extrapolation. In the more general Romberg case, we can use Neville’s
algorithm (see §3.1) to extrapolate the successive refinements to zero stepsize.
Neville’s algorithm can in fact be coded very concisely within a Romberg integration
routine. For clarity of the program, however, it seems better to do the extrapolation
by subroutine call to polint, already given in §3.1.

SUBROUTINE qromb(func,a,b,ss)
INTEGER JMAX,JMAXP,K,KM
REAL a,b,func,ss,EPS
EXTERNAL func
PARAMETER (EPS=1.e-6, JMAX=20, JMAXP=JMAX+1, K=5, KM=K-1)

C USES polint,trapzd
Returns as ss the integral of the function func from a to b. Integration is performed by
Romberg’s method of order 2K, where, e.g., K=2 is Simpson’s rule.
Parameters: EPS is the fractional accuracy desired, as determined by the extrapolation
error estimate; JMAX limits the total number of steps; K is the number of points used in
the extrapolation.

INTEGER j
REAL dss,h(JMAXP),s(JMAXP) These store the successive trapezoidal approximations

and their relative stepsizes.h(1)=1.
do 11 j=1,JMAX

call trapzd(func,a,b,s(j),j)
if (j.ge.K) then

call polint(h(j-KM),s(j-KM),K,0.,ss,dss)
if (abs(dss).le.EPS*abs(ss)) return

endif
s(j+1)=s(j)
h(j+1)=0.25*h(j) This is a key step: The factor is 0.25 even though

the stepsize is decreased by only 0.5. This makes
the extrapolation a polynomial in h2 as allowed
by equation (4.2.1), not just a polynomial in h.

enddo 11

pause ’too many steps in qromb’
END

The routine qromb, along with its required trapzd and polint, is quite
powerful for sufficiently smooth (e.g., analytic) integrands, integrated over intervals
which contain no singularities, and where the endpoints are also nonsingular. qromb,
in such circumstances, takes many, many fewer function evaluations than either of
the routines in §4.2. For example, the integral

∫ 2

0

x4 log(x +
√

x2 + 1)dx

converges (with parameters as shown above) on the very first extrapolation, after
just 5 calls totrapzd, whileqsimp requires 8 calls (8 times as many evaluations of
the integrand) andqtrap requires 13 calls (making 256 times as many evaluations
of the integrand).

CITED REFERENCES AND FURTHER READING:

Stoer, J., and Bulirsch, R. 1980, Introduction to Numerical Analysis (New York: Springer-Verlag),
§§3.4–3.5.

Dahlquist, G., and Bjorck, A. 1974, Numerical Methods (Englewood Cliffs, NJ: Prentice-Hall),
§§7.4.1–7.4.2.

Ralston, A., and Rabinowitz, P. 1978, A First Course in Numerical Analysis, 2nd ed. (New York:
McGraw-Hill), §4.10–2.

4.4 Improper Integrals

For our present purposes, an integral will be “improper” if it has any of the
following problems:

• its integrand goes to a finite limiting value at finite upper and lower limits,
but cannot be evaluatedright on one of those limits (e.g.,sin x/x atx = 0)

• its upper limit is∞ , or its lower limit is−∞
• it has an integrable singularity at either limit (e.g.,x−1/2 at x = 0)
• it has an integrable singularity at a known place between its upper and

lower limits
• it has an integrable singularity at an unknown place between its upper

and lower limits
If an integral is infinite (e.g.,

∫ ∞
1 x−1dx), or does not exist in a limiting sense

(e.g.,
∫ ∞
−∞ cosxdx), we do not call it improper; we call it impossible. No amount of

clever algorithmics will return a meaningful answer to an ill-posed problem.
In this section we will generalize the techniques of the preceding two sections

to cover the first four problems on the above list. A more advanced discussion of
quadrature with integrable singularities occurs in Chapter 18, notably§18.3. The
fifth problem, singularity at unknown location, can really only be handled by the
use of a variable stepsize differential equation integration routine, as will be given
in Chapter 16.

We need a workhorse like the extended trapezoidal rule (equation 4.1.11), but
one which is anopen formula in the sense of§4.1, i.e., does not require the integrand
to be evaluated at the endpoints. Equation (4.1.19), the extended midpoint rule, is the
best choice. The reason is that (4.1.19) shares with (4.1.11) the “deep” property of
having an error series that is entirely even inh. Indeed there is a formula, not as well
known as it ought to be, called theSecond Euler-Maclaurin summation formula,∫ xN

x1

f(x)dx = h[f3/2 + f5/2 + f7/2 + · · · + fN−3/2 + fN−1/2]

+
B2h

2

4
(f ′

N − f ′
1) + · · ·

+
B2kh2k

(2k)!
(1 − 2−2k+1)(f (2k−1)

N − f
(2k−1)
1 ) + · · ·

(4.4.1)

This equation can be derived by writing out (4.2.1) with stepsize h, then writing it
out again with stepsize h/2, then subtracting the first from twice the second.

It is not possible to double the number of steps in the extended midpoint rule
and still have the benefit of previous function evaluations (try it!). However, it is
possible to triple the number of steps and do so. Shall we do this, or double and
accept the loss? On the average, tripling does a factor

√
3 of unnecessary work,

since the “right” number of steps for a desired accuracy criterion may in fact fall
anywhere in the logarithmic interval implied by tripling. For doubling, the factor
is only

√
2, but we lose an extra factor of 2 in being unable to use all the previous

evaluations. Since 1.732 < 2 × 1.414, it is better to triple.
Here is the resulting routine, which is directly comparable to trapzd.

SUBROUTINE midpnt(func,a,b,s,n)
INTEGER n
REAL a,b,s,func
EXTERNAL func

This routine computes the nth stage of refinement of an extended midpoint rule. func is
input as the name of the function to be integrated between limits a and b, also input. When

called with n=1, the routine returns as s the crudest estimate of
∫ b
a f(x)dx. Subsequent

calls with n=2,3,... (in that sequential order) will improve the accuracy of s by adding

(2/3) × 3n-1 additional interior points. s should not be modified between sequential calls.
INTEGER it,j
REAL ddel,del,sum,tnm,x
if (n.eq.1) then

s=(b-a)*func(0.5*(a+b))
else

it=3**(n-2)
tnm=it
del=(b-a)/(3.*tnm)
ddel=del+del The added points alternate in spacing between del and ddel.
x=a+0.5*del
sum=0.
do 11 j=1,it

sum=sum+func(x)
x=x+ddel
sum=sum+func(x)
x=x+del

enddo 11

s=(s+(b-a)*sum/tnm)/3. The new sum is combined with the old integral to give a
refined integral.endif

return
END

The routine midpnt can exactly replace trapzd in a driver routine like qtrap
(§4.2); one simply changes call trapzd to call midpnt, and perhaps also
decreases the parameter JMAX since 3JMAX−1 (from step tripling) is a much larger
number than 2JMAX−1 (step doubling).

The open formula implementation analogous to Simpson’s rule (qsimp in §4.2)
substitutes midpnt for trapzd and decreases JMAX as above, but now also changes
the extrapolation step to be

s=(9.*st-ost)/8.

since, when the number of steps is tripled, the error decreases to 1/9th its size, not
1/4th as with step doubling.

Either the modified qtrap or the modified qsimp will fix the first problem
on the list at the beginning of this section. Yet more sophisticated is to generalize
Romberg integration in like manner:

SUBROUTINE qromo(func,a,b,ss,choose)
INTEGER JMAX,JMAXP,K,KM
REAL a,b,func,ss,EPS
EXTERNAL func,choose
PARAMETER (EPS=1.e-6, JMAX=14, JMAXP=JMAX+1, K=5, KM=K-1)

C USES polint
Romberg integration on an open interval. Returns as ss the integral of the function func
from a to b, using any specified integrating subroutine choose and Romberg’s method.
Normally choose will be an open formula, not evaluating the function at the endpoints. It
is assumed that choose triples the number of steps on each call, and that its error series
contains only even powers of the number of steps. The routines midpnt, midinf, midsql,
midsqu, are possible choices for choose. The parameters have the same meaning as in
qromb.

INTEGER j
REAL dss,h(JMAXP),s(JMAXP)
h(1)=1.
do 11 j=1,JMAX

call choose(func,a,b,s(j),j)
if (j.ge.K) then

call polint(h(j-KM),s(j-KM),K,0.,ss,dss)
if (abs(dss).le.EPS*abs(ss)) return

endif
s(j+1)=s(j)
h(j+1)=h(j)/9. This is where the assumption of step tripling and an even

error series is used.enddo 11

pause ’too many steps in qromo’
END

The differences between qromo and qromb (§4.3) are so slight that it is perhaps
gratuitous to list qromo in full. It, however, is an excellent driver routine for solving
all the other problems of improper integrals in our first list (except the intractable
fifth), as we shall now see.

The basic trick for improper integrals is to make a change of variables to
eliminate the singularity, or to map an infinite range of integration to a finite one.
For example, the identity

∫ b

a

f(x)dx =
∫ 1/a

1/b

1
t2

f

(
1
t

)
dt ab > 0 (4.4.2)

can be used with either b → ∞ and a positive, or with a → −∞ and b negative, and
works for any function which decreases towards infinity faster than 1/x 2.

You can make the change of variable implied by (4.4.2) either analytically and
then use (e.g.) qromo and midpnt to do the numerical evaluation, or you can let
the numerical algorithm make the change of variable for you. We prefer the latter
method as being more transparent to the user. To implement equation (4.4.2) we
simply write a modified version of midpnt, called midinf, which allows b to be
infinite (or, more precisely, a very large number on your particular machine, such
as 1 × 1030), or a to be negative and infinite.

SUBROUTINE midinf(funk,aa,bb,s,n)
INTEGER n
REAL aa,bb,s,funk
EXTERNAL funk

This routine is an exact replacement for midpnt, i.e., returns as s the nth stage of refinement
of the integral of funk from aa to bb, except that the function is evaluated at evenly spaced
points in 1/x rather than in x. This allows the upper limit bb to be as large and positive
as the computer allows, or the lower limit aa to be as large and negative, but not both.
aa and bb must have the same sign.

INTEGER it,j
REAL a,b,ddel,del,sum,tnm,func,x
func(x)=funk(1./x)/x**2 This statement function effects the change of variable.
b=1./aa These two statements change the limits of integration ac-

cordingly.a=1./bb
if (n.eq.1) then From this point on, the routine is exactly identical to midpnt.

s=(b-a)*func(0.5*(a+b))
else

it=3**(n-2)
tnm=it
del=(b-a)/(3.*tnm)
ddel=del+del
x=a+0.5*del
sum=0.
do 11 j=1,it

sum=sum+func(x)
x=x+ddel
sum=sum+func(x)
x=x+del

enddo 11

s=(s+(b-a)*sum/tnm)/3.
endif
return
END

If you need to integrate from a negative lower limit to positive infinity, you do
this by breaking the integral into two pieces at some positive value, for example,

call qromo(funk,-5.,2.,s1,midpnt)
call qromo(funk,2.,1.e30,s2,midinf)
answer=s1+s2

Where should you choose the breakpoint? At a sufficiently large positive value so
that the function funk is at least beginning to approach its asymptotic decrease to
zero value at infinity. The polynomial extrapolation implicit in the second call to
qromo deals with a polynomial in 1/x, not in x.

To deal with an integral that has an integrable power-law singularity at its lower
limit, one also makes a change of variable. If the integrand diverges as (x − a)−γ ,
0 ≤ γ < 1, near x = a, use the identity

∫ b

a

f(x)dx =
1

1 − γ

∫ (b−a)1−γ

0

t
γ

1−γ f(t
1

1−γ + a)dt (b > a) (4.4.3)

If the singularity is at the upper limit, use the identity

∫ b

a

f(x)dx =
1

1 − γ

∫ (b−a)1−γ

0

t
γ

1−γ f(b − t
1

1−γ )dt (b > a) (4.4.4)

If there is a singularity at both limits, divide the integral at an interior breakpoint
as in the example above.

Equations (4.4.3) and (4.4.4) are particularly simple in the case of inverse
square-root singularities, a case that occurs frequently in practice:

∫ b

a

f(x)dx =
∫ √

b−a

0

2tf(a + t2)dt (b > a) (4.4.5)

for a singularity at a, and

∫ b

a

f(x)dx =
∫ √

b−a

0

2tf(b − t2)dt (b > a) (4.4.6)

for a singularity at b. Once again, we can implement these changes of variable
transparently to the user by defining substitute routines for midpnt which make the
change of variable automatically:

SUBROUTINE midsql(funk,aa,bb,s,n)
INTEGER n
REAL aa,bb,s,funk
EXTERNAL funk

This routine is an exact replacement for midpnt, except that it allows for an inverse square-
root singularity in the integrand at the lower limit aa.

INTEGER it,j
REAL ddel,del,sum,tnm,x,func,a,b
func(x)=2.*x*funk(aa+x**2)
b=sqrt(bb-aa)
a=0.
if (n.eq.1) then

The rest of the routine is exactly like midpnt and is omitted.

Similarly,

SUBROUTINE midsqu(funk,aa,bb,s,n)
INTEGER n
REAL aa,bb,s,funk
EXTERNAL funk

This routine is an exact replacement for midpnt, except that it allows for an inverse square-
root singularity in the integrand at the upper limit bb.

INTEGER it,j
REAL ddel,del,sum,tnm,x,func,a,b
func(x)=2.*x*funk(bb-x**2)
b=sqrt(bb-aa)
a=0.
if (n.eq.1) then

The rest of the routine is exactly like midpnt and is omitted.

One last example should suffice to show how these formulas are derived in
general. Suppose the upper limit of integration is infinite, and the integrand falls off
exponentially. Then we want a change of variable that maps e−xdx into (±)dt (with
the sign chosen to keep the upper limit of the new variable larger than the lower
limit). Doing the integration gives by inspection

t = e−x or x = − log t (4.4.7)

so that

∫ x=∞

x=a

f(x)dx =
∫ t=e−a

t=0

f(− log t)
dt

t
(4.4.8)

The user-transparent implementation would be

SUBROUTINE midexp(funk,aa,bb,s,n)
INTEGER n
REAL aa,bb,s,funk
EXTERNAL funk

This routine is an exact replacement for midpnt, except that bb is assumed to be infinite
(value passed not actually used). It is assumed that the function funk decreases exponen-
tially rapidly at infinity.

INTEGER it,j
REAL ddel,del,sum,tnm,x,func,a,b
func(x)=funk(-log(x))/x
b=exp(-aa)
a=0.
if (n.eq.1) then

The rest of the routine is exactly like midpnt and is omitted.

CITED REFERENCES AND FURTHER READING:

Acton, F.S. 1970, Numerical Methods That Work; 1990, corrected edition (Washington: Mathe-
matical Association of America), Chapter 4.

Dahlquist, G., and Bjorck, A. 1974, Numerical Methods (Englewood Cliffs, NJ: Prentice-Hall),
§7.4.3, p. 294.

Stoer, J., and Bulirsch, R. 1980, Introduction to Numerical Analysis (New York: Springer-Verlag),
§3.7, p. 152.

4.5 Gaussian Quadratures and Orthogonal
Polynomials

In the formulas of §4.1, the integral of a function was approximated by the sum
of its functional values at a set of equally spaced points, multiplied by certain aptly
chosen weighting coefficients. We saw that as we allowed ourselves more freedom
in choosing the coefficients, we could achieve integration formulas of higher and
higher order. The idea of Gaussian quadratures is to give ourselves the freedom to
choose not only the weighting coefficients, but also the location of the abscissas at
which the function is to be evaluated: They will no longer be equally spaced. Thus,
we will have twice the number of degrees of freedom at our disposal; it will turn out
that we can achieve Gaussian quadrature formulas whose order is, essentially, twice
that of the Newton-Cotes formula with the same number of function evaluations.

Does this sound too good to be true? Well, in a sense it is. The catch is a
familiar one, which cannot be overemphasized: High order is not the same as high
accuracy. High order translates to high accuracy only when the integrand is very
smooth, in the sense of being “well-approximated by a polynomial.”



4.5 Gaussian Quadratures and Orthogonal Polynomials 141

S
am

ple page from
 N

U
M

E
R

IC
A

L R
E

C
IP

E
S

 IN
 F

O
R

T
R

A
N

 77: T
H

E
 A

R
T

 O
F

 S
C

IE
N

T
IF

IC
 C

O
M

P
U

T
IN

G
 (IS

B
N

 0-521-43064-X
)

C
opyright (C

) 1986-1992 by C
am

bridge U
niversity P

ress.P
rogram

s C
opyright (C

) 1986-1992 by N
um

erical R
ecipes S

oftw
are. 

P
erm

ission is granted for internet users to m
ake one paper copy for their ow

n personal use. F
urther reproduction, or any copying of m

achine-
readable files (including this one) to any server

com
puter, is strictly prohibited. T

o order N
um

erical R
ecipes books

or C
D

R
O

M
s, visit w

ebsite
http://w

w
w

.nr.com
 or call 1-800-872-7423 (N

orth A
m

erica only),or send em
ail to directcustserv@

cam
bridge.org (outside N

orth A
m

erica).

There is, however, one additional feature of Gaussian quadrature formulas that
adds to their usefulness: We can arrange the choice of weights and abscissas to make
the integral exact for a class of integrands “polynomials times some known function
W (x)” rather than for the usual class of integrands “polynomials.” The function
W (x) can then be chosen to remove integrable singularities from the desired integral.
Given W (x), in other words, and given an integer N , we can find a set of weights
wj and abscissas xj such that the approximation

∫ b

a

W (x)f(x)dx ≈
N∑

j=1

wjf(xj) (4.5.1)

is exact if f(x) is a polynomial. For example, to do the integral∫ 1

−1

exp(− cos2 x)√
1 − x2

dx (4.5.2)

(not a very natural looking integral, it must be admitted), we might well be interested
in a Gaussian quadrature formula based on the choice

W (x) =
1√

1 − x2
(4.5.3)

in the interval (−1, 1). (This particular choice is called Gauss-Chebyshev integration,
for reasons that will become clear shortly.)

Notice that the integration formula (4.5.1) can also be written with the weight
function W (x) not overtly visible: Define g(x) ≡ W (x)f(x) and vj ≡ wj/W (xj).
Then (4.5.1) becomes

∫ b

a

g(x)dx ≈
N∑

j=1

vjg(xj) (4.5.4)

Where did the function W (x) go? It is lurking there, ready to give high-order
accuracy to integrands of the form polynomials times W (x), and ready to deny high-
order accuracy to integrands that are otherwise perfectly smooth and well-behaved.
When you find tabulations of the weights and abscissas for a given W (x), you have
to determine carefully whether they are to be used with a formula in the form of
(4.5.1), or like (4.5.4).

Here is an example of a quadrature routine that contains the tabulated abscissas
and weights for the case W (x) = 1 and N = 10. Since the weights and abscissas
are, in this case, symmetric around the midpoint of the range of integration, there
are actually only five distinct values of each:

SUBROUTINE qgaus(func,a,b,ss)
REAL a,b,ss,func
EXTERNAL func

Returns as ss the integral of the function func between a and b, by ten-point Gauss-
Legendre integration: the function is evaluated exactly ten times at interior points in the
range of integration.

INTEGER j

REAL dx,xm,xr,w(5),x(5) The abscissas and weights.
SAVE w,x
DATA w/.2955242247,.2692667193,.2190863625,.1494513491,.0666713443/
DATA x/.1488743389,.4333953941,.6794095682,.8650633666,.9739065285/
xm=0.5*(b+a)
xr=0.5*(b-a)
ss=0 Will be twice the average value of the function, since the ten

weights (five numbers above each used twice) sum to 2.do 11 j=1,5
dx=xr*x(j)
ss=ss+w(j)*(func(xm+dx)+func(xm-dx))

enddo 11

ss=xr*ss Scale the answer to the range of integration.
return
END

The above routine illustrates that one can use Gaussian quadratures without
necessarily understanding the theory behind them: One just locates tabulated weights
and abscissas in a book (e.g., [1] or [2]). However, the theory is very pretty, and it
will come in handy if you ever need to construct your own tabulation of weights and
abscissas for an unusual choice of W (x). We will therefore give, without any proofs,
some useful results that will enable you to do this. Several of the results assume that
W (x) does not change sign inside (a, b), which is usually the case in practice.

The theory behind Gaussian quadratures goes back to Gauss in 1814, who
used continued fractions to develop the subject. In 1826 Jacobi rederived Gauss’s
results by means of orthogonal polynomials. The systematic treatment of arbitrary
weight functions W (x) using orthogonal polynomials is largely due to Christoffel in
1877. To introduce these orthogonal polynomials, let us fix the interval of interest
to be (a, b). We can define the “scalar product of two functions f and g over a
weight function W ” as

〈f |g〉 ≡
∫ b

a

W (x)f(x)g(x)dx (4.5.5)

The scalar product is a number, not a function of x. Two functions are said to be
orthogonal if their scalar product is zero. A function is said to be normalized if its
scalar product with itself is unity. A set of functions that are all mutually orthogonal
and also all individually normalized is called an orthonormal set.

We can find a set of polynomials (i) that includes exactly one polynomial of
order j, called pj(x), for each j = 0, 1, 2, . . . , and (ii) all of which are mutually
orthogonal over the specified weight function W (x). A constructive procedure for
finding such a set is the recurrence relation

p−1(x) ≡ 0

p0(x) ≡ 1

pj+1(x) = (x − aj)pj(x) − bjpj−1(x) j = 0, 1, 2, . . .

(4.5.6)

where

aj =
〈xpj |pj〉
〈pj |pj〉 j = 0, 1, . . .

bj =
〈pj|pj〉

〈pj−1|pj−1〉 j = 1, 2, . . .

(4.5.7)

The coefficient b0 is arbitrary; we can take it to be zero.
The polynomials defined by (4.5.6) are monic, i.e., the coefficient of their

leading term [xj for pj(x)] is unity. If we divide each pj(x) by the constant
[〈pj |pj〉]1/2 we can render the set of polynomials orthonormal. One also encounters
orthogonal polynomials with various other normalizations. You can convert from
a given normalization to monic polynomials if you know that the coefficient of
xj in pj is λj , say; then the monic polynomials are obtained by dividing each p j

by λj . Note that the coefficients in the recurrence relation (4.5.6) depend on the
adopted normalization.

The polynomial pj(x) can be shown to have exactly j distinct roots in the
interval (a, b). Moreover, it can be shown that the roots of p j(x) “interleave” the
j − 1 roots of pj−1(x), i.e., there is exactly one root of the former in between each
two adjacent roots of the latter. This fact comes in handy if you need to find all the
roots: You can start with the one root of p1(x) and then, in turn, bracket the roots
of each higher j, pinning them down at each stage more precisely by Newton’s rule
or some other root-finding scheme (see Chapter 9).

Why would you ever want to find all the roots of an orthogonal polynomial
pj(x)? Because the abscissas of the N -point Gaussian quadrature formulas (4.5.1)
and (4.5.4) with weighting function W (x) in the interval (a, b) are precisely the roots
of the orthogonal polynomial pN(x) for the same interval and weighting function.
This is the fundamental theorem of Gaussian quadratures, and lets you find the
abscissas for any particular case.

Once you know the abscissas x1, . . . , xN , you need to find the weights wj ,
j = 1, . . . , N . One way to do this (not the most efficient) is to solve the set of
linear equations

p0(x1) . . . p0(xN )
p1(x1) . . . p1(xN )

...
...

pN−1(x1) . . . pN−1(xN )




w1

w2
...

wN

 =


∫ b

a W (x)p0(x)dx
0
...
0

 (4.5.8)

Equation (4.5.8) simply solves for those weights such that the quadrature (4.5.1)
gives the correct answer for the integral of the first N orthogonal polynomials. Note
that the zeros on the right-hand side of (4.5.8) appear because p 1(x), . . . , pN−1(x)
are all orthogonal to p0(x), which is a constant. It can be shown that, with those
weights, the integral of the next N−1 polynomials is also exact, so that the quadrature
is exact for all polynomials of degree 2N − 1 or less. Another way to evaluate the
weights (though one whose proof is beyond our scope) is by the formula

wj =
〈pN−1|pN−1〉

pN−1(xj)p′N (xj)
(4.5.9)

where p′N (xj) is the derivative of the orthogonal polynomial at its zero x j .
The computation of Gaussian quadrature rules thus involves two distinct phases:

(i) the generation of the orthogonal polynomials p 0, . . . , pN , i.e., the computation of
the coefficients aj , bj in (4.5.6); (ii) the determination of the zeros of pN (x), and
the computation of the associated weights. For the case of the “classical” orthogonal
polynomials, the coefficients aj and bj are explicitly known (equations 4.5.10 –

4.5.14 below) and phase (i) can be omitted. However, if you are confronted with a
“nonclassical” weight function W (x), and you don’t know the coefficients a j and
bj , the construction of the associated set of orthogonal polynomials is not trivial.
We discuss it at the end of this section.

Computation of the Abscissas and Weights

This task can range from easy to difficult, depending on how much you already
know about your weight function and its associated polynomials. In the case of
classical, well-studied, orthogonal polynomials, practically everything is known,
including good approximations for their zeros. These can be used as starting guesses,
enabling Newton’s method (to be discussed in §9.4) to converge very rapidly.
Newton’s method requires the derivative p ′

N (x), which is evaluated by standard
relations in terms of pN and pN−1. The weights are then conveniently evaluated by
equation (4.5.9). For the following named cases, this direct root-finding is faster,
by a factor of 3 to 5, than any other method.

Here are the weight functions, intervals, and recurrence relations that generate
the most commonly used orthogonal polynomials and their corresponding Gaussian
quadrature formulas.

Gauss-Legendre:

W (x) = 1 − 1 < x < 1

(j + 1)Pj+1 = (2j + 1)xPj − jPj−1 (4.5.10)

Gauss-Chebyshev:

W (x) = (1 − x2)−1/2 − 1 < x < 1

Tj+1 = 2xTj − Tj−1 (4.5.11)

Gauss-Laguerre:

W (x) = xαe−x 0 < x < ∞

(j + 1)Lα
j+1 = (−x + 2j + α + 1)Lα

j − (j + α)Lα
j−1 (4.5.12)

Gauss-Hermite:

W (x) = e−x2 −∞ < x < ∞

Hj+1 = 2xHj − 2jHj−1 (4.5.13)

Gauss-Jacobi:

W (x) = (1 − x)α(1 + x)β − 1 < x < 1

cjP
(α,β)
j+1 = (dj + ejx)P (α,β)

j − fjP
(α,β)
j−1 (4.5.14)

where the coefficients cj , dj , ej , and fj are given by

cj = 2(j + 1)(j + α + β + 1)(2j + α + β)

dj = (2j + α + β + 1)(α2 − β2)

ej = (2j + α + β)(2j + α + β + 1)(2j + α + β + 2)

fj = 2(j + α)(j + β)(2j + α + β + 2)

(4.5.15)

We now give individual routines that calculate the abscissas and weights for
these cases. First comes the most common set of abscissas and weights, those of
Gauss-Legendre. The routine, due to G.B. Rybicki, uses equation (4.5.9) in the
special form for the Gauss-Legendre case,

wj =
2

(1 − x2
j )[P

′
N (xj)]2

(4.5.16)

The routine also scales the range of integration from (x1, x2) to (−1, 1), and provides
abscissas xj and weights wj for the Gaussian formula

∫ x2

x1

f(x)dx =
N∑

j=1

wjf(xj) (4.5.17)

SUBROUTINE gauleg(x1,x2,x,w,n)
INTEGER n
REAL x1,x2,x(n),w(n)
DOUBLE PRECISION EPS
PARAMETER (EPS=3.d-14) EPS is the relative precision.

Given the lower and upper limits of integration x1 and x2, and given n, this routine returns
arrays x(1:n) and w(1:n) of length n, containing the abscissas and weights of the Gauss-
Legendre n-point quadrature formula.

INTEGER i,j,m
DOUBLE PRECISION p1,p2,p3,pp,xl,xm,z,z1

High precision is a good idea for this routine.
m=(n+1)/2 The roots are symmetric in the interval, so we

only have to find half of them.xm=0.5d0*(x2+x1)
xl=0.5d0*(x2-x1)
do 12 i=1,m Loop over the desired roots.

z=cos(3.141592654d0*(i-.25d0)/(n+.5d0))
Starting with the above approximation to the ith root, we enter the main loop of re-
finement by Newton’s method.

1 continue
p1=1.d0
p2=0.d0
do 11 j=1,n Loop up the recurrence relation to get the Leg-

endre polynomial evaluated at z.p3=p2
p2=p1
p1=((2.d0*j-1.d0)*z*p2-(j-1.d0)*p3)/j

enddo 11

p1 is now the desired Legendre polynomial. We next compute pp, its derivative, by
a standard relation involving also p2, the polynomial of one lower order.

pp=n*(z*p1-p2)/(z*z-1.d0)

z1=z
z=z1-p1/pp Newton’s method.

if(abs(z-z1).gt.EPS)goto 1
x(i)=xm-xl*z Scale the root to the desired interval,
x(n+1-i)=xm+xl*z and put in its symmetric counterpart.
w(i)=2.d0*xl/((1.d0-z*z)*pp*pp) Compute the weight
w(n+1-i)=w(i) and its symmetric counterpart.

enddo 12

return
END

Next we give three routines that use initial approximations for the roots given
by Stroud and Secrest [2]. The first is for Gauss-Laguerre abscissas and weights, to
be used with the integration formula∫ ∞

0

xαe−xf(x)dx =
N∑

j=1

wjf(xj) (4.5.18)

SUBROUTINE gaulag(x,w,n,alf)
INTEGER n,MAXIT
REAL alf,w(n),x(n)
DOUBLE PRECISION EPS
PARAMETER (EPS=3.D-14,MAXIT=10) Increase EPS if you don’t have this precision.

C USES gammln
Given alf, the parameter α of the Laguerre polynomials, this routine returns arrays x(1:n)
and w(1:n) containing the abscissas and weights of the n-point Gauss-Laguerre quadrature
formula. The smallest abscissa is returned in x(1), the largest in x(n).

INTEGER i,its,j
REAL ai,gammln
DOUBLE PRECISION p1,p2,p3,pp,z,z1
High precision is a good idea for this routine.

do 13 i=1,n Loop over the desired roots.
if(i.eq.1)then Initial guess for the smallest root.

z=(1.+alf)*(3.+.92*alf)/(1.+2.4*n+1.8*alf)
else if(i.eq.2)then Initial guess for the second root.

z=z+(15.+6.25*alf)/(1.+.9*alf+2.5*n)
else Initial guess for the other roots.

ai=i-2
z=z+((1.+2.55*ai)/(1.9*ai)+1.26*ai*alf/

* (1.+3.5*ai))*(z-x(i-2))/(1.+.3*alf)
endif
do 12 its=1,MAXIT Refinement by Newton’s method.

p1=1.d0
p2=0.d0
do 11 j=1,n Loop up the recurrence relation to get the Laguerre

polynomial evaluated at z.p3=p2
p2=p1
p1=((2*j-1+alf-z)*p2-(j-1+alf)*p3)/j

enddo 11

p1 is now the desired Laguerre polynomial. We next compute pp, its derivative, by
a standard relation involving also p2, the polynomial of one lower order.

pp=(n*p1-(n+alf)*p2)/z
z1=z
z=z1-p1/pp Newton’s formula.
if(abs(z-z1).le.EPS)goto 1

enddo 12

pause ’too many iterations in gaulag’
1 x(i)=z Store the root and the weight.



4.5 Gaussian Quadratures and Orthogonal Polynomials 147

S
am

ple page from
 N

U
M

E
R

IC
A

L R
E

C
IP

E
S

 IN
 F

O
R

T
R

A
N

 77: T
H

E
 A

R
T

 O
F

 S
C

IE
N

T
IF

IC
 C

O
M

P
U

T
IN

G
 (IS

B
N

 0-521-43064-X
)

C
opyright (C

) 1986-1992 by C
am

bridge U
niversity P

ress.P
rogram

s C
opyright (C

) 1986-1992 by N
um

erical R
ecipes S

oftw
are. 

P
erm

ission is granted for internet users to m
ake one paper copy for their ow

n personal use. F
urther reproduction, or any copying of m

achine-
readable files (including this one) to any server

com
puter, is strictly prohibited. T

o order N
um

erical R
ecipes books

or C
D

R
O

M
s, visit w

ebsite
http://w

w
w

.nr.com
 or call 1-800-872-7423 (N

orth A
m

erica only),or send em
ail to directcustserv@

cam
bridge.org (outside N

orth A
m

erica).

w(i)=-exp(gammln(alf+n)-gammln(float(n)))/(pp*n*p2)
enddo 13

return
END

Next is a routine for Gauss-Hermite abscissas and weights. If we use the
“standard” normalization of these functions, as given in equation (4.5.13), we find
that the computations overflow for large N because of various factorials that occur.
We can avoid this by using instead the orthonormal set of polynomials H̃j . They
are generated by the recurrence

H̃−1 = 0, H̃0 =
1

π1/4
, H̃j+1 = x

√
2

j + 1
H̃j −

√
j

j + 1
H̃j−1 (4.5.19)

The formula for the weights becomes

wj =
2

[H̃ ′
N (xj)]2

(4.5.20)

while the formula for the derivative with this normalization is

H̃ ′
j =

√
2jH̃j−1 (4.5.21)

The abscissas and weights returned by gauher are used with the integration formula

∫ ∞

−∞
e−x2

f(x)dx =
N∑

j=1

wjf(xj) (4.5.22)

SUBROUTINE gauher(x,w,n)
INTEGER n,MAXIT
REAL w(n),x(n)
DOUBLE PRECISION EPS,PIM4
PARAMETER (EPS=3.D-14,PIM4=.7511255444649425D0,MAXIT=10)

Given n, this routine returns arrays x(1:n) and w(1:n) containing the abscissas and
weights of the n-point Gauss-Hermite quadrature formula. The largest abscissa is returned
in x(1), the most negative in x(n).
Parameters: EPS is the relative precision, PIM4 = 1/π1/4, MAXIT = maximum iterations.

INTEGER i,its,j,m
DOUBLE PRECISION p1,p2,p3,pp,z,z1
High precision is a good idea for this routine.

m=(n+1)/2
The roots are symmetric about the origin, so we have to find only half of them.

do 13 i=1,m Loop over the desired roots.
if(i.eq.1)then Initial guess for the largest root.

z=sqrt(float(2*n+1))-1.85575*(2*n+1)**(-.16667)
else if(i.eq.2)then Initial guess for the second largest root.

z=z-1.14*n**.426/z
else if (i.eq.3)then Initial guess for the third largest root.

z=1.86*z-.86*x(1)
else if (i.eq.4)then Initial guess for the fourth largest root.

z=1.91*z-.91*x(2)
else Initial guess for the other roots.

z=2.*z-x(i-2)

endif
do 12 its=1,MAXIT Refinement by Newton’s method.

p1=PIM4
p2=0.d0
do 11 j=1,n Loop up the recurrence relation to get the Hermite poly-

nomial evaluated at z.p3=p2
p2=p1
p1=z*sqrt(2.d0/j)*p2-sqrt(dble(j-1)/dble(j))*p3

enddo 11

p1 is now the desired Hermite polynomial. We next compute pp, its derivative, by
the relation (4.5.21) using p2, the polynomial of one lower order.

pp=sqrt(2.d0*n)*p2
z1=z
z=z1-p1/pp Newton’s formula.
if(abs(z-z1).le.EPS)goto 1

enddo 12

pause ’too many iterations in gauher’
1 x(i)=z Store the root

x(n+1-i)=-z and its symmetric counterpart.
w(i)=2.d0/(pp*pp) Compute the weight
w(n+1-i)=w(i) and its symmetric counterpart.

enddo 13

return
END

Finally, here is a routine for Gauss-Jacobi abscissas and weights, which
implement the integration formula

∫ 1

−1

(1 − x)α(1 + x)βf(x)dx =
N∑

j=1

wjf(xj) (4.5.23)

SUBROUTINE gaujac(x,w,n,alf,bet)
INTEGER n,MAXIT
REAL alf,bet,w(n),x(n)
DOUBLE PRECISION EPS
PARAMETER (EPS=3.D-14,MAXIT=10) Increase EPS if you don’t have this precision.

C USES gammln
Given alf and bet, the parameters α and β of the Jacobi polynomials, this routine returns
arrays x(1:n) and w(1:n) containing the abscissas and weights of the n-point Gauss-Jacobi
quadrature formula. The largest abscissa is returned in x(1), the smallest in x(n).

INTEGER i,its,j
REAL alfbet,an,bn,r1,r2,r3,gammln
DOUBLE PRECISION a,b,c,p1,p2,p3,pp,temp,z,z1
High precision is a good idea for this routine.

do 13 i=1,n Loop over the desired roots.
if(i.eq.1)then Initial guess for the largest root.

an=alf/n
bn=bet/n
r1=(1.+alf)*(2.78/(4.+n*n)+.768*an/n)
r2=1.+1.48*an+.96*bn+.452*an*an+.83*an*bn
z=1.-r1/r2

else if(i.eq.2)then Initial guess for the second largest root.
r1=(4.1+alf)/((1.+alf)*(1.+.156*alf))
r2=1.+.06*(n-8.)*(1.+.12*alf)/n
r3=1.+.012*bet*(1.+.25*abs(alf))/n
z=z-(1.-z)*r1*r2*r3

else if(i.eq.3)then Initial guess for the third largest root.
r1=(1.67+.28*alf)/(1.+.37*alf)
r2=1.+.22*(n-8.)/n

r3=1.+8.*bet/((6.28+bet)*n*n)
z=z-(x(1)-z)*r1*r2*r3

else if(i.eq.n-1)then Initial guess for the second smallest root.
r1=(1.+.235*bet)/(.766+.119*bet)
r2=1./(1.+.639*(n-4.)/(1.+.71*(n-4.)))
r3=1./(1.+20.*alf/((7.5+alf)*n*n))
z=z+(z-x(n-3))*r1*r2*r3

else if(i.eq.n)then Initial guess for the smallest root.
r1=(1.+.37*bet)/(1.67+.28*bet)
r2=1./(1.+.22*(n-8.)/n)
r3=1./(1.+8.*alf/((6.28+alf)*n*n))
z=z+(z-x(n-2))*r1*r2*r3

else Initial guess for the other roots.
z=3.*x(i-1)-3.*x(i-2)+x(i-3)

endif
alfbet=alf+bet
do 12 its=1,MAXIT Refinement by Newton’s method.

temp=2.d0+alfbet Start the recurrence with P0 and P1 to avoid a divi-
sion by zero when α + β = 0 or −1.p1=(alf-bet+temp*z)/2.d0

p2=1.d0
do 11 j=2,n Loop up the recurrence relation to get the Jacobi

polynomial evaluated at z.p3=p2
p2=p1
temp=2*j+alfbet
a=2*j*(j+alfbet)*(temp-2.d0)
b=(temp-1.d0)*(alf*alf-bet*bet+temp*

* (temp-2.d0)*z)
c=2.d0*(j-1+alf)*(j-1+bet)*temp
p1=(b*p2-c*p3)/a

enddo 11

pp=(n*(alf-bet-temp*z)*p1+2.d0*(n+alf)*
* (n+bet)*p2)/(temp*(1.d0-z*z))

p1 is now the desired Jacobi polynomial. We next compute pp, its derivative, by a
standard relation involving also p2, the polynomial of one lower order.

z1=z
z=z1-p1/pp Newton’s formula.
if(abs(z-z1).le.EPS)goto 1

enddo 12

pause ’too many iterations in gaujac’
1 x(i)=z Store the root and the weight.

w(i)=exp(gammln(alf+n)+gammln(bet+n)-gammln(n+1.)-
* gammln(n+alfbet+1.))*temp*2.**alfbet/(pp*p2)

enddo 13

return
END

Legendre polynomials are special cases of Jacobi polynomials with α = β = 0,
but it is worth having the separate routine for them, gauleg, given above. Chebyshev
polynomials correspond to α = β = −1/2 (see §5.8). They have analytic abscissas
and weights:

xj = cos
(

π(j − 1
2 )

N

)
wj =

π

N

(4.5.24)

Case of Known Recurrences

Turn now to the case where you do not know good initial guesses for the zeros of your
orthogonal polynomials, but you do have available the coefficients aj and bj that generate
them. As we have seen, the zeros of pN(x) are the abscissas for the N -point Gaussian
quadrature formula. The most useful computational formula for the weights is equation
(4.5.9) above, since the derivative p′N can be efficiently computed by the derivative of (4.5.6)
in the general case, or by special relations for the classical polynomials. Note that (4.5.9) is
valid as written only for monic polynomials; for other normalizations, there is an extra factor
of λN/λN−1, where λN is the coefficient of xN in pN .

Except in those special cases already discussed, the best way to find the abscissas is not
to use a root-finding method like Newton’s method on pN(x). Rather, it is generally faster
to use the Golub-Welsch [3] algorithm, which is based on a result of Wilf [4]. This algorithm
notes that if you bring the term xpj to the left-hand side of (4.5.6) and the term pj+1 to the
right-hand side, the recurrence relation can be written in matrix form as

x


p0

p1

...
pN−2

pN−1

 =


a0 1
b1 a1 1

...
...

bN−2 aN−2 1
bN−1 aN−1

 ·


p0

p1

...
pN−2

pN−1

 +


0
0
...
0

pN


or

xp = T · p + pN eN−1 (4.5.25)

Here T is a tridiagonal matrix, p is a column vector of p0, p1, . . . , pN−1, and eN−1 is a unit
vector with a 1 in the (N − 1)st (last) position and zeros elsewhere. The matrix T can be
symmetrized by a diagonal similarity transformation D to give

J = DTD−1 =


a0

√
b1√

b1 a1

√
b2

...
...√

bN−2 aN−2

√
bN−1√

bN−1 aN−1

 (4.5.26)

The matrix J is called the Jacobi matrix (not to be confused with other matrices named
after Jacobi that arise in completely different problems!). Now we see from (4.5.25) that
pN(xj) = 0 is equivalent to xj being an eigenvalue of T. Since eigenvalues are preserved
by a similarity transformation, xj is an eigenvalue of the symmetric tridiagonal matrix J.
Moreover, Wilf [4] shows that if vj is the eigenvector corresponding to the eigenvalue xj ,
normalized so that v · v = 1, then

wj = µ0v
2
j,1 (4.5.27)

where

µ0 =

∫ b

a

W (x)dx (4.5.28)

and where vj,1 is the first component of v. As we shall see in Chapter 11, finding all
eigenvalues and eigenvectors of a symmetric tridiagonal matrix is a relatively efficient and
well-conditioned procedure. We accordingly give a routine, gaucof, for finding the abscissas
and weights, given the coefficients aj and bj . Remember that if you know the recurrence
relation for orthogonal polynomials that are not normalized to be monic, you can easily convert
it to monic form by means of the quantities λj .

SUBROUTINE gaucof(n,a,b,amu0,x,w)
INTEGER n,NMAX
REAL amu0,a(n),b(n),w(n),x(n)
PARAMETER (NMAX=64)

C USES eigsrt,tqli
Computes the abscissas and weights for a Gaussian quadrature formula from the Jacobi
matrix. On input, a(1:n) and b(1:n) are the coefficients of the recurrence relation for

the set of monic orthogonal polynomials. The quantity µ0 ≡ ∫ b
a W (x)dx is input as amu0.

The abscissas x(1:n) are returned in descending order, with the corresponding weights
in w(1:n). The arrays a and b are modified. Execution can be speeded up by modifying
tqli and eigsrt to compute only the first component of each eigenvector.

INTEGER i,j
REAL z(NMAX,NMAX)
do 12 i=1,n

if(i.ne.1)b(i)=sqrt(b(i)) Set up superdiagonal of Jacobi matrix.
do 11 j=1,n Set up identity matrix for tqli to compute eigenvectors.

if(i.eq.j)then
z(i,j)=1.

else
z(i,j)=0.

endif
enddo 11

enddo 12

call tqli(a,b,n,NMAX,z)
call eigsrt(a,z,n,NMAX) Sort eigenvalues into descending order.
do 13 i=1,n

x(i)=a(i)
w(i)=amu0*z(1,i)**2 Equation (4.5.12).

enddo 13

return
END

Orthogonal Polynomials with Nonclassical Weights

This somewhat specialized subsection will tell you what to do if your weight function
is not one of the classical ones dealt with above and you do not know the aj’s and bj ’s
of the recurrence relation (4.5.6) to use in gaucof. Then, a method of finding the aj ’s
and bj ’s is needed.

The procedure of Stieltjes is to compute a0 from (4.5.7), then p1(x) from (4.5.6).
Knowing p0 and p1, we can compute a1 and b1 from (4.5.7), and so on. But how are we
to compute the inner products in (4.5.7)?

The textbook approach is to represent each pj(x) explicitly as a polynomial in x and
to compute the inner products by multiplying out term by term. This will be feasible if we
know the first 2N moments of the weight function,

µj =

∫ b

a

xjW (x)dx j = 0, 1, . . . , 2N − 1 (4.5.29)

However, the solution of the resulting set of algebraic equations for the coefficients aj and bj

in terms of the moments µj is in general extremely ill-conditioned. Even in double precision,
it is not unusual to lose all accuracy by the time N = 12. We thus reject any procedure
based on the moments (4.5.29).

Sack and Donovan [5] discovered that the numerical stability is greatly improved if,
instead of using powers of x as a set of basis functions to represent the pj’s, one uses some
other known set of orthogonal polynomials πj(x), say. Roughly speaking, the improved
stability occurs because the polynomial basis “samples” the interval (a, b) better than the
power basis when the inner product integrals are evaluated, especially if its weight function
resembles W (x).

So assume that we know the modified moments

νj =

∫ b

a

πj(x)W (x)dx j = 0, 1, . . . , 2N − 1 (4.5.30)

where the πj ’s satisfy a recurrence relation analogous to (4.5.6),

π−1(x) ≡ 0

π0(x) ≡ 1

πj+1(x) = (x − αj)πj(x) − βjπj−1(x) j = 0, 1, 2, . . .

(4.5.31)

and the coefficients αj , βj are known explicitly. Then Wheeler [6] has given an efficient
O(N2) algorithm equivalent to that of Sack and Donovan for finding aj and bj via a set
of intermediate quantities

σk,l = 〈pk|πl〉 k, l ≥ −1 (4.5.32)

Initialize

σ−1,l = 0 l = 1, 2, . . . , 2N − 2

σ0,l = νl l = 0, 1, . . . , 2N − 1

a0 = α0 +
ν1

ν0

b0 = 0

(4.5.33)

Then, for k = 1, 2, . . . , N − 1, compute

σk,l = σk−1,l+1 − (ak−1 − αl)σk−1,l − bk−1σk−2,l + βlσk−1,l−1

l = k, k + 1, . . . , 2N − k − 1

ak = αk − σk−1,k

σk−1,k−1
+

σk,k+1

σk,k

bk =
σk,k

σk−1,k−1

(4.5.34)

Note that the normalization factors can also easily be computed if needed:

〈p0|p0〉 = ν0

〈pj |pj〉 = bj 〈pj−1|pj−1〉 j = 1, 2, . . .
(4.5.35)

You can find a derivation of the above algorithm in Ref. [7].
Wheeler’s algorithm requires that the modified moments (4.5.30) be accurately computed.

In practical cases there is often a closed form, or else recurrence relations can be used. The
algorithm is extremely successful for finite intervals (a, b). For infinite intervals, the algorithm
does not completely remove the ill-conditioning. In this case, Gautschi [8,9] recommends
reducing the interval to a finite interval by a change of variable, and then using a suitable
discretization procedure to compute the inner products. You will have to consult the
references for details.

We give the routine orthog for generating the coefficients aj and bj by Wheeler’s
algorithm, given the coefficients αj and βj , and the modified moments νj . To conform to
the usual FORTRAN convention for dimensioning subscripts, the indices of the σ matrix are
increased by 2, i.e., sig(k,l) = σk−2,l−2, while the indices of the vectors α, β, a and
b are increased by 1.

SUBROUTINE orthog(n,anu,alpha,beta,a,b)
INTEGER n,NMAX
REAL a(n),alpha(2*n-1),anu(2*n),b(n),beta(2*n-1)
PARAMETER (NMAX=64)

Computes the coefficients aj and bj , j = 0, . . . N − 1, of the recurrence relation for
monic orthogonal polynomials with weight function W (x) by Wheeler’s algorithm. On input,
alpha(1:2*n-1) and beta(1:2*n-1) are the coefficients αj and βj , j = 0, . . . 2N − 2,
of the recurrence relation for the chosen basis of orthogonal polynomials. The modified
moments νj are input in anu(1:2*n). The first n coefficients are returned in a(1:n) and
b(1:n).

INTEGER k,l
REAL sig(2*NMAX+1,2*NMAX+1)
do 11 l=3,2*n Initialization, Equation (4.5.33).

sig(1,l)=0.
enddo 11

do 12 l=2,2*n+1
sig(2,l)=anu(l-1)

enddo 12

a(1)=alpha(1)+anu(2)/anu(1)
b(1)=0.
do 14 k=3,n+1 Equation (4.5.34).

do 13 l=k,2*n-k+3
sig(k,l)=sig(k-1,l+1)+(alpha(l-1)-a(k-2))*sig(k-1,l)-

* b(k-2)*sig(k-2,l)+beta(l-1)*sig(k-1,l-1)
enddo 13

a(k-1)=alpha(k-1)+sig(k,k+1)/sig(k,k)-sig(k-1,k)/sig(k-1,k-1)
b(k-1)=sig(k,k)/sig(k-1,k-1)

enddo 14

return
END

As an example of the use of orthog, consider the problem [7] of generating orthogonal
polynomials with the weight function W (x) = − log x on the interval (0, 1). A suitable set
of πj ’s is the shifted Legendre polynomials

πj =
(j!)2

(2j)!
Pj(2x − 1) (4.5.36)

The factor in front of Pj makes the polynomials monic. The coefficients in the recurrence
relation (4.5.31) are

αj =
1

2
j = 0, 1, . . .

βj =
1

4(4 − j−2)
j = 1, 2, . . .

(4.5.37)

while the modified moments are

νj =


1 j = 0
(−1)j(j!)2

j(j + 1)(2j)!
j ≥ 1

(4.5.38)

A call to orthog with this input allows one to generate the required polynomials to machine
accuracy for very large N , and hence do Gaussian quadrature with this weight function. Before
Sack and Donovan’s observation, this seemingly simple problem was essentially intractable.

Extensions of Gaussian Quadrature

There are many different ways in which the ideas of Gaussian quadrature have
been extended. One important extension is the case of preassigned nodes: Some
points are required to be included in the set of abscissas, and the problem is to choose

the weights and the remaining abscissas to maximize the degree of exactness of the
the quadrature rule. The most common cases are Gauss-Radau quadrature, where
one of the nodes is an endpoint of the interval, either a or b, and Gauss-Lobatto
quadrature, where both a and b are nodes. Golub [10] has given an algorithm similar
to gaucof for these cases.

The second important extension is the Gauss-Kronrod formulas. For ordinary
Gaussian quadrature formulas, as N increases the sets of abscissas have no points
in common. This means that if you compare results with increasing N as a way of
estimating the quadrature error, you cannot reuse the previous function evaluations.
Kronrod [11] posed the problem of searching for optimal sequences of rules, each
of which reuses all abscissas of its predecessor. If one starts with N = m, say,
and then adds n new points, one has 2n + m free parameters: the n new abscissas
and weights, and m new weights for the fixed previous abscissas. The maximum
degree of exactness one would expect to achieve would therefore be 2n + m − 1.
The question is whether this maximum degree of exactness can actually be achieved
in practice, when the abscissas are required to all lie inside (a, b). The answer to
this question is not known in general.

Kronrod showed that if you choose n = m + 1, an optimal extension can
be found for Gauss-Legendre quadrature. Patterson [12] showed how to compute
continued extensions of this kind. Sequences such as N = 10, 21, 43, 87, . . . are
popular in automatic quadrature routines [13] that attempt to integrate a function until
some specified accuracy has been achieved.

CITED REFERENCES AND FURTHER READING:

Abramowitz, M., and Stegun, I.A. 1964, Handbook of Mathematical Functions, Applied Mathe-
matics Series, Volume 55 (Washington: National Bureau of Standards; reprinted 1968 by
Dover Publications, New York), §25.4. [1]

Stroud, A.H., and Secrest, D. 1966, Gaussian Quadrature Formulas (Englewood Cliffs, NJ:
Prentice-Hall). [2]

Golub, G.H., and Welsch, J.H. 1969, Mathematics of Computation, vol. 23, pp. 221–230 and
A1–A10. [3]

Wilf, H.S. 1962, Mathematics for the Physical Sciences (New York: Wiley), Problem 9, p. 80. [4]

Sack, R.A., and Donovan, A.F. 1971/72, Numerische Mathematik, vol. 18, pp. 465–478. [5]

Wheeler, J.C. 1974, Rocky Mountain Journal of Mathematics, vol. 4, pp. 287–296. [6]

Gautschi, W. 1978, in Recent Advances in Numerical Analysis, C. de Boor and G.H. Golub, eds.
(New York: Academic Press), pp. 45–72. [7]

Gautschi, W. 1981, in E.B. Christoffel, P.L. Butzer and F. Fehér, eds. (Basel: Birkhauser Verlag),
pp. 72–147. [8]

Gautschi, W. 1990, in Orthogonal Polynomials, P. Nevai, ed. (Dordrecht: Kluwer Academic Pub-
lishers), pp. 181–216. [9]

Golub, G.H. 1973, SIAM Review, vol. 15, pp. 318–334. [10]

Kronrod, A.S. 1964, Doklady Akademii Nauk SSSR, vol. 154, pp. 283–286 (in Russian). [11]

Patterson, T.N.L. 1968, Mathematics of Computation, vol. 22, pp. 847–856 and C1–C11; 1969,
op. cit., vol. 23, p. 892. [12]

Piessens, R., de Doncker, E., Uberhuber, C.W., and Kahaner, D.K. 1983, QUADPACK: A Sub-
routine Package for Automatic Integration (New York: Springer-Verlag). [13]

Stoer, J., and Bulirsch, R. 1980, Introduction to Numerical Analysis (New York: Springer-Verlag),
§3.6.

Johnson, L.W., and Riess, R.D. 1982, Numerical Analysis, 2nd ed. (Reading, MA: Addison-
Wesley), §6.5.



4.6 Multidimensional Integrals 155

S
am

ple page from
 N

U
M

E
R

IC
A

L R
E

C
IP

E
S

 IN
 F

O
R

T
R

A
N

 77: T
H

E
 A

R
T

 O
F

 S
C

IE
N

T
IF

IC
 C

O
M

P
U

T
IN

G
 (IS

B
N

 0-521-43064-X
)

C
opyright (C

) 1986-1992 by C
am

bridge U
niversity P

ress.P
rogram

s C
opyright (C

) 1986-1992 by N
um

erical R
ecipes S

oftw
are. 

P
erm

ission is granted for internet users to m
ake one paper copy for their ow

n personal use. F
urther reproduction, or any copying of m

achine-
readable files (including this one) to any server

com
puter, is strictly prohibited. T

o order N
um

erical R
ecipes books

or C
D

R
O

M
s, visit w

ebsite
http://w

w
w

.nr.com
 or call 1-800-872-7423 (N

orth A
m

erica only),or send em
ail to directcustserv@

cam
bridge.org (outside N

orth A
m

erica).

Carnahan, B., Luther, H.A., and Wilkes, J.O. 1969, Applied Numerical Methods (New York:
Wiley), §§2.9–2.10.

Ralston, A., and Rabinowitz, P. 1978, A First Course in Numerical Analysis, 2nd ed. (New York:
McGraw-Hill), §§4.4–4.8.

4.6 Multidimensional Integrals

Integrals of functions of several variables, over regions with dimension greater
than one, are not easy. There are two reasons for this. First, the number of function
evaluations needed to sample an N -dimensional space increases as the N th power
of the number needed to do a one-dimensional integral. If you need 30 function
evaluations to do a one-dimensional integral crudely, then you will likely need on
the order of 30000 evaluations to reach the same crude level for a three-dimensional
integral. Second, the region of integration in N -dimensional space is defined by
an N − 1 dimensional boundary which can itself be terribly complicated: It need
not be convex or simply connected, for example. By contrast, the boundary of a
one-dimensional integral consists of two numbers, its upper and lower limits.

The first question to be asked, when faced with a multidimensional integral,
is, “can it be reduced analytically to a lower dimensionality?” For example,
so-called iterated integrals of a function of one variable f(t) can be reduced to
one-dimensional integrals by the formula∫ x

0

dtn

∫ tn

0

dtn−1 · · ·
∫ t3

0

dt2

∫ t2

0

f(t1)dt1

=
1

(n − 1)!

∫ x

0

(x − t)n−1f(t)dt

(4.6.1)

Alternatively, the function may have some special symmetry in the way it depends
on its independent variables. If the boundary also has this symmetry, then the
dimension can be reduced. In three dimensions, for example, the integration of a
spherically symmetric function over a spherical region reduces, in polar coordinates,
to a one-dimensional integral.

The next questions to be asked will guide your choice between two entirely
different approaches to doing the problem. The questions are: Is the shape of the
boundary of the region of integration simple or complicated? Inside the region, is
the integrand smooth and simple, or complicated, or locally strongly peaked? Does
the problem require high accuracy, or does it require an answer accurate only to
a percent, or a few percent?

If your answers are that the boundary is complicated, the integrand is not
strongly peaked in very small regions, and relatively low accuracy is tolerable, then
your problem is a good candidate for Monte Carlo integration. This method is very
straightforward to program, in its cruder forms. One needs only to know a region
with simple boundaries that includes the complicated region of integration, plus a
method of determining whether a random point is inside or outside the region of
integration. Monte Carlo integration evaluates the function at a random sample of

points, and estimates its integral based on that random sample. We will discuss it in
more detail, and with more sophistication, in Chapter 7.

If the boundary is simple, and the function is very smooth, then the remaining
approaches, breaking up the problem into repeated one-dimensional integrals, or
multidimensional Gaussian quadratures, will be effective and relatively fast [1]. If
you require high accuracy, these approaches are in any case the only ones available
to you, since Monte Carlo methods are by nature asymptotically slow to converge.

For low accuracy, use repeated one-dimensional integration or multidimensional
Gaussian quadratures when the integrand is slowly varying and smooth in the region
of integration, Monte Carlo when the integrand is oscillatory or discontinuous, but
not strongly peaked in small regions.

If the integrand is strongly peaked in small regions, and you know where those
regions are, break the integral up into several regions so that the integrand is smooth
in each, and do each separately. If you don’t know where the strongly peaked regions
are, you might as well (at the level of sophistication of this book) quit: It is hopeless
to expect an integration routine to search out unknown pockets of large contribution
in a huge N -dimensional space. (But see §7.8.)

If, on the basis of the above guidelines, you decide to pursue the repeated one-
dimensional integration approach, here is how it works. For definiteness, we will
consider the case of a three-dimensional integral in x, y, z-space. Two dimensions,
or more than three dimensions, are entirely analogous.

The first step is to specify the region of integration by (i) its lower and upper
limits in x, which we will denote x1 and x2; (ii) its lower and upper limits in y at
a specified value of x, denoted y1(x) and y2(x); and (iii) its lower and upper limits
in z at specified x and y, denoted z1(x, y) and z2(x, y). In other words, find the
numbers x1 and x2, and the functions y1(x), y2(x), z1(x, y), and z2(x, y) such that

I ≡
∫ ∫ ∫

dx dy dzf(x, y, z)

=
∫ x2

x1

dx

∫ y2(x)

y1(x)

dy

∫ z2(x,y)

z1(x,y)

dz f(x, y, z)
(4.6.2)

For example, a two-dimensional integral over a circle of radius one centered on
the origin becomes

∫ 1

−1

dx

∫ √
1−x2

−√
1−x2

dy f(x, y) (4.6.3)

Now we can define a function G(x, y) that does the innermost integral,

G(x, y) ≡
∫ z2(x,y)

z1(x,y)

f(x, y, z)dz (4.6.4)

and a function H(x) that does the integral of G(x, y),

H(x) ≡
∫ y2(x)

y1(x)

G(x, y)dy (4.6.5)

Figure 4.6.1. Function evaluations for a two-dimensional integral over an irregular region, shown
schematically. The outer integration routine, in y, requests values of the inner, x, integral at locations
along the y axis of its own choosing. The inner integration routine then evaluates the function at
x locations suitable to it. This is more accurate in general than, e.g., evaluating the function on a
Cartesian mesh of points.

and finally our answer as an integral over H(x)

I =
∫ x2

x1

H(x)dx (4.6.6)

To implement equations (4.6.4)–(4.6.6) in a program, one needs three separate
copies of a basic one-dimensional integration routine (and of any subroutines called
by it), one each for the x, y, and z integrations. If you try to make do with only
one copy, then it will call itself recursively, since (e.g.) the function evaluations
of H for the x integration will themselves call the integration routine to do the y
integration (see Figure 4.6.1). In our example, let us suppose that we plan to use the
one-dimensional integrator qgaus of §4.5. Then we make three identical copies and
call them qgausx, qgausy, and qgausz. The basic program for three-dimensional
integration then is as follows:

SUBROUTINE quad3d(x1,x2,ss)
REAL ss,x1,x2,h
EXTERNAL h

C USES h,qgausx
Returns as ss the integral of a user-supplied function func over a three-dimensional region
specified by the limits x1, x2, and by the user-supplied functions y1, y2, z1, and z2, as
defined in (4.6.2).

call qgausx(h,x1,x2,ss)
return
END

FUNCTION f(zz)
REAL f,zz,func,x,y,z

COMMON /xyz/ x,y,z
C USES func

Called by qgausz. Calls func.
z=zz
f=func(x,y,z)
return
END

FUNCTION g(yy)
REAL g,yy,f,z1,z2,x,y,z
EXTERNAL f
COMMON /xyz/ x,y,z

C USES f,qgausz,z1,z2
Called by qgausy. Calls qgausz.

REAL ss
y=yy
call qgausz(f,z1(x,y),z2(x,y),ss)
g=ss
return
END

FUNCTION h(xx)
REAL h,xx,g,y1,y2,x,y,z
EXTERNAL g
COMMON /xyz/ x,y,z

C USES g,qgausy,y1,y2
Called by qgausx. Calls qgausy.

REAL ss
x=xx
call qgausy(g,y1(x),y2(x),ss)
h=ss
return
END

The necessary user-supplied functions have the following calling sequences:

FUNCTION func(x,y,z) The 3-dimensional function to be integrated
FUNCTION y1(x)
FUNCTION y2(x)
FUNCTION z1(x,y)
FUNCTION z2(x,y)

